Creating TI Distribution Table by Land Use Type
June 12, 2006

Bongghi Hong

Previous document describes how the TI map (ti10m) can be obtained from DEM:

This document describes how the TI distribution can be obtained for each land use type. As an example, we will use the NLCD land use map available on the HBEF website (http://www.hubbardbrook.org/research/gis/gis.htm). Download the “Hubbard Brook National Land Cover Dataset 1992” (nlcd_92.exe), extract the “nlcd_92.e00” interchange file, use ArcToolbox to convert it to coverage file, and add the map to ArcMap:

Let’s clip the land use map using the TI map. Follow these steps: (1) Click on “Spatial Analyst > Option…” and specify the analysis mask as “ti10m” in the “General” tab. (2) Click on “Extent” tab and specify the analysis extent as “Same as layer ti10m”. (3) Click on “Cell Size” tab and specify the cell size as “Same as layer ti10m”. (4) Click “OK” and click on “Spatial Analyst > Raster Calculator…” (5) Double-click “nlcd_92” and click “Evaluate”. (6) Right-click the name of resulting raster map (on the left “Display” window), choose “Make Permanent…”, and save the map as “lu10m”:

Now we will create a multi-band raster map containing both “ti10m” and “lu10m”. Open the Raster Calculator again and type the following equation:
combine([ti10m],[lu10m])
Click “Evaluate” and make permanent (save) as “combined10m”:

Right-click the “combined10m” and select “Open Attribute Table”:

[image: image5.png]e ol

Objectn | Value Count Titom Lutom
v 0 1 == 5 2
7 2 850 6 2
2 3 2 7 2
3 4 29274 4 5
4 5 22287 G @
5 6 S 4 2
6 7 1434 s 2
7 s e 5 5
s s #137 7 @
s 0 652 3 2
0 1 Ed 3 5
1 12 %12 s 2
12 1 00 s 5
1 14 EiE 5 [
14 15 26845 6 a
15 1 152 5 £
1 7 103 7 £
7 18 w20 s a
18 18 L 4 a
[53 [

Record 14 _||— _| _| Show [ 81 Selected | Fecads (0 out of 181 Selected) Opiions -





Click “Options > Export…” and save the table as “Combined10m.txt” (text file format). You can open this file using Excel:

Save the text file as Excel format (Combined10m.xls). Open the Visual Basic Editor (“Tools > Macro > Visual Basic Editor…”) and paste the following code as “CreateTIDistribution” procedure:
Sub CreateTIDistribution()

    ' Specify Input Sheet Name

    InputSheetName = "Combined10m"

    Sheets(InputSheetName).Activate

    ' Obtain Column Heading, Number of Rows, Minimum and Maximum Values

    TIName = Cells(1, 4)

    LUName = Cells(1, 5)

    RowCount = Sheets(InputSheetName).UsedRange.Rows.Count

    TIRange = Range(Cells(2, 4), Cells(RowCount, 4))

    LURange = Range(Cells(2, 5), Cells(RowCount, 5))

    MinTI = Application.WorksheetFunction.Min(TIRange)

    MaxTI = Application.WorksheetFunction.Max(TIRange)

    MinLU = Application.WorksheetFunction.Min(LURange)

    MaxLU = Application.WorksheetFunction.Max(LURange)

    ' Add New Sheet for TI Distribution Table

    Worksheets.Add Before:=Worksheets(1)

    Sheets(1).Activate

    ' Set Up the TI Distribution Table

    Cells(2, 1) = TIName

    Cells(1, 2) = LUName

    CountTIRow = 2

    For i = MinTI To MaxTI

        CountTIRow = CountTIRow + 1

        Cells(CountTIRow, 1) = i

    Next i

    CountLUCol = 1

    For i = MinLU To MaxLU

        Set FindLU = Sheets(InputSheetName).Range( _

            Sheets(InputSheetName).Cells(2, 5), _

            Sheets(InputSheetName).Cells(RowCount, 5)) _

            .Find(i, LookAt:=xlWhole)

        If Not FindLU Is Nothing Then

            CountLUCol = CountLUCol + 1

            Cells(2, CountLUCol) = i

        End If

    Next i

    Range(Cells(1, 1), Cells(CountTIRow, 1)).Font.Bold = True

    Range(Cells(1, 1), Cells(2, CountLUCol)).Font.Bold = True

    Range(Cells(1, 1), Cells(CountTIRow, 1)).Font.Color = RGB(0, 0, 255)

    Range(Cells(1, 1), Cells(2, CountLUCol)).Font.Color = RGB(0, 0, 255)

    ' Fill the TI Distribution Table with Zeros Values

    For i = 3 To CountTIRow

        For j = 2 To CountLUCol

            Cells(i, j) = 0

        Next j

    Next i

    ' Read the Count Values and Write to the Table

    For i = 2 To RowCount

        GetCellCount = Sheets(InputSheetName).Cells(i, 3)

        GetTI = Sheets(InputSheetName).Cells(i, 4)

        GetLU = Sheets(InputSheetName).Cells(i, 5)

        Set FindTI = Range(Cells(3, 1), Cells(CountTIRow, 1)). _

            Find(GetTI, LookAt:=xlWhole)

        Set FindLU = Range(Cells(2, 2), Cells(2, CountLUCol)). _

            Find(GetLU, LookAt:=xlWhole)

        Cells(FindTI.Row, FindLU.Column) = GetCellCount

    Next i

End Sub
After running the Macro, you will get the following table reporting TI distributions by land use tpye:

